
INTRODUCTION

This application note describes a unique method for
implementing interrupts in software on the PIC16C5X
series of microcontrollers. The method takes advantage
of the PIC16C5X’s architecture which allows changing
the program counter under software control. Up to eight
interrupt lines are possible, but the practical limit for
simple code generation is six interrupts, or 64 possible
input conditions. The interrupt detection time is under
software control and standard I/O pins are used as the
interrupt lines.

THEORY OF OPERATION

SOFTWARE POLLING OF I/O LINES REPLACES
HARDWARE INTERRUPT

The interrupt conditions are determined by detecting
changes on the I/O lines that  have been selected to be
the interrupt lines. These changes are used to create a
jump table that allows a different program response to
each interrupt condition. The interrupt response time is
under software control and can be as short as ten to
twenty microseconds, depending on  main program and
interrupt subroutine program length.

CREATING  THE INTERRUPT SUBROUTINE JUMP
TABLE

Each I/O condition may have its own unique subroutine
to respond to changes on the interrupt lines.  Direct
access to these routines is achieved by using the
PIC16C5X’s  ability to change the program counter
under software control. Here is an example of how two
I/O lines may be polled:

MOVF CONDTN,W ;LOAD I/0 CONDITION INTO W
;REGISTER

ANDLW 3 ;MASK OFF TOP 6 BITS

ADDWF 2,1 ;ADD INPUT TO PROGRAM COUNTER

;TO CREATE JUMP TABLE

GOTO MAIN ;FOR  NO CHANGE GO TO MAIN
;PROGRAM

GOTO INT1 ;FOR CHANGE IN BIT 0 GOTO INT1

GOTO INT2 ;FOR CHANGE IN BIT 1 GOTO INT2

GOTO INT3 ;FOR BOTH CHANGE GOTO INT3

The changes to the I/O lines have been used to create
a two bit number that is added to the program counter.
The GOTO that is executed depends on the new pro-
gram counter address.

CREATING CONSTANT TIME
POLLING

In most applications requiring interrupts, it is important to
poll the interrupt lines at fixed time intervals, usually only
a few microseconds in length. Two techniques may be
used on the PIC16C5X to achieve this. They are dividing
the main program into multiple sections and implement-
ing an elapsed time counter (see flow chart).  Both of
these techniques use the same program jump table
concept that was described above. First, the main pro-
gram is divided into several sections based on the
desired I/O polling time. When MAIN is called a branch
register is added to the program counter. This deter-
mines which section of MAIN code should be executed
next. At the end of execution the branch register is
decremented so the next section of code will be ex-
ecuted after the next polling. If the branch register is zero
then the number of sections of main code is added into
it to start the main program over again.

An elapsed time counter can be implemented using the
RTCC counter. At the beginning of I/O polling the RTCC
register is cleared. It then starts counting the instruction
cycles. Then after the main program subsection has
been executed, the RTCC register is subtracted from the
desired polling time. This determines how many instruc-
tions need to be executed before the next polling. A jump
table is then created to execute these instructions before
the next polling.  An example is shown below. This
example assumes  from zero to 15 additional instruction
cycles are needed. Actual numbers need to be com-
puted for each individual application.

MOVLW POLL ;POLL=DESIRED POLL CYCLES - 15

SUBWF RTCC,W ;DETERMINE HOW MUCH TIME TO WAIT

ADDWF 2,1 ;ADD WAIT  TIME TO PROGRAM
COUNTER

NOP ;15 ADDITIONAL INSTRUCTION CYCLES

:

: ;TOTAL OF 15 NOP’S

NOP ;1 ADDITIONAL INSTRUCTION CYCLES

GOTO START ;0 ADDITIONAL INSTRUCTION CYCLES

FIGURE 1 - SOFTWARE INTERRUPT FLOW CHART

For example, if the desired instruction time is 50 cycles
and the subsection we just executed has a consumed a
total of 40 instruction cycles (including all overhead
cycles) the value of

RTCC(40) - POLL(50-15(35)) =5

will  be added to the program counter. The program will
then jump to the sixth NOP. That NOP plus the 9
following it will be executed for a total of ten more
instruction cycles. Note that the final GOTO has two

instruction cycles and these must be included in the
program overhead.

Example

The following example (see flow chart and code) is the
core program for the software interrupt technique de-
scribed above. This program assumes four interrupt
conditions, four main program sections and an eight
additional elapsed time instructions.

MPASM B0.54                                                             PAGE  1

                            LIST    P=16C54

                                            ;SOFTWARE INTERRUPT APPLICATIONS
                                            ;BRANCH IS MAIN PROGRAM REGISTER
0008                 BRANCH EQU     8
0009                 CNDTN  EQU     9
000A                 IO     EQU     0A
000B                 TEMP   EQU     0B

0000 0069            SETUP  CLRF    CNDTN
0001 0C04                   MOVLW   4
0002 0028                   MOVWF   BRANCH  ;FOUR MAIN PROGRAM SECTIONS
0003 0C08                   MOVLW   8
0004 0002                   OPTION          ;SET RTCC TO ONE COUNT PER INSTRUCTION CYCLE

0005 0061            START  CLRF    1       ;CLEAR RTCC REGISTER
0006 0206                   MOVF    6,W     ;READ I/O
0007 002A                   MOVWF   IO
0008 0109                   IORWF   CNDTN,W ;THIS SECTION OF CODE CALCULATES THE
0009 002B                   MOVWF   TEMP    ;JUMP TABLE. ANY INPUT THAT CHANGES FROM
000A 0209                   MOVF    CNDTN,W ;A ZERO TO A ONE IS CONSIDERED AN INTERRUPT.
000B 00AB                   SUBWF   TEMP,1  ;THE EQUATION IS:
000C 020A                   MOVF    IO,W    ;  (IO + CNDTN) - CNDTN = INTERRUPT
000D 0029                   MOVWF   CNDTN   ;WHERE IO IS CURRENT INPUT AND
000E 020B                   MOVF    TEMP,W  ;CNDTN IS PREVIOUS INPUT.
000F 0E03                   ANDLW   3       ;MASK OFF TOP 6 BITS
0010 01E2                   ADDWF   2,1     ;ADD INPUT TO PC TO CREATE JUMP TABLE
0011 0A1B                   GOTO    MAIN    ;FOR INPUT=00
0012 0A15                   GOTO    INT1    ;FOR INPUT=01
0013 0A17                   GOTO    INT2    ;FOR INPUT=10
0014 0A19                   GOTO    INT3    ;FOR INPUT=11

0015 0000            INT1   NOP             ;INTERRUPT LINE 1 CODE
0016 0A05                   GOTO    START
0017 0000            INT2   NOP             ;INTERRUPT LINE 2 CODE
0018 0A05                   GOTO    START
0019 0000            INT3   NOP             ;INTERRUPT LINES 1 AND 2 CODE
001A 0A05                   GOTO    START

001B 0208            MAIN   MOVF    BRANCH,W
001C 01E2                   ADDWF   2,1     ;ADD BRANCH TO PC TO CREATE JUMP TABLE
001D 0000                   NOP
001E 0A28                   GOTO    MAIN4   ;JUMP TABLE, LAST FIRST ON DECREMENT TABLE
001F 0A26                   GOTO    MAIN3
0020 0A24                   GOTO    MAIN2
0021 0A22                   GOTO    MAIN1

0022 0000            MAIN1  NOP             ;MAIN PROGRAM CODE BANK ONE
0023 0A2A                   GOTO    BRNCHK
0024 0000            MAIN2  NOP             ;MAIN PROGRAM CODE SECTION TWO
0025 0A2A                   GOTO    BRNCHK
0026 0000            MAIN3  NOP             ;MAIN PROGRAM CODE SECTION THREE
0027 0A2A                   GOTO    BRNCHK
0028 0000            MAIN4  NOP             ;MAIN PROGRAM CODE SECTION FOUR
0029 0A2A                   GOTO    BRNCHK

002A 02E8            BRNCHK DECFSZ  BRANCH,1 ;DECREMENT BRANCH REGISTER AND CHECK FOR ZERO
002B 0A2E                   GOTO    TIMCHK
002C 0C04                   MOVLW   4
002D 0028                   MOVWF   BRANCH  ;RELOAD BRANCH WITH 4 AT END OF MAIN

002E 0C29            TIMCHK MOVLW   D’41'   ;CHECK TO SEE IF RTCC HAS REACHED 50(50-7)

APPENDIX A:

002F 0081                   SUBWF   1,W     ;DETERMINE WAIT TIME
0030 01E2                   ADDWF   2,1     ;ADD WAIT TIME TO PC
0031 0000                   NOP
0032 0000                   NOP
0033 0000                   NOP
0034 0000                   NOP
0035 0000                   NOP
0036 0000                   NOP
0037 0000                   NOP
0038 0A05                   GOTO    START
                            END

Errors   :    0
Warnings :    0

